
PRAKTIKUM 9 

PROSEDUR BERSARANG 

 

1. MINGGU KE : 11 

2. PERALATAN : LCD, E-LEARNING 

3. SOFTWARE : MAPLE 

4. TUJUAN 

 Mahasiswa memahami: 

 Prosedur bersarang 

 Variabel lokal dan variable global 

 Operasi partisi  

 Algoritma quick sort 

 

5. TEORI PENGANTAR DAN LANGKAH KERJA 

Suatu prosedur Maple dapat didefinisikan di dalam procedure Maple yang lain. 

Perintah map untuk menerapkan suatu operasi pada setiap unsur dari suatu tipe struktur. 

Misalnya untuk membagi setiap unsur dari suatu list dengan suatu bilangan 

 

> lst:=[8,4,2,16]: 

> map(x->x/8,lst); 

Perintah map juga dapat digunakan dalam prosedur. Prosedur berikut adalah untuk 

membagi setiap unsur dari list dengan unsur pertama dari list tersebut. 

 

> nest:=proc(x::list) 

>   local v; 

>   v:=x[1]; 

>   map(y->y/v,x); 

> end proc: 

> nest(lst); 

Dalam prosedur ini Maple menggunakan map sebagai contoh dari prosedur bersarang 

yang menyatakan v dalam map sebagai v yang sama dalam prosedur yang lebih luar, 

yaitu nest. 

 

A. Variabel Lokal dan Variabel Global 

 

 Selanjutnya akan dijelaskan bagaimana Maple menggunakan variabel lokal dan 

variabel global. Ketika membuat prosedur, sebaiknya dinyatakan secara eksplisit yang 

mana variabel lokal dan mana variabel global. Dalam prosedur nest di atas, variable 

dalam perintah map mengambil nilai dari prosedur yang mengelilinginya. Bagaimana jika 

variabel v didefinisikan sebagai variabel lokal? 

 

> nest2:=proc(x::list) 

>   local v; 

>   v:=x[1]; 

>   map(proc(y) local v; y/v; end, x); 

> end proc: 

> nest2(lst); 

B. Algoritma QuickSort 

Algoritma pengurutan adalah hal yang menarik dalam ilmu komputer. Algoritma quick 

sort berikut adalah algoritma klasik. Kunci untuk memahami algoritma ini adalah dengan 

memahami operasi dari partisi. Teknik pengurutannya adalah mengambil satu bilangan 

dari array yang akan diurutkan. Lalu me-reposisi bilangan tadi  

 Ambil satu bilangan dari array yang akan diurutkan. 

 Reposisi bilangan-bilangan dalam array yang kurang dari bilangan yang dipilih 

tadi di satu sisi array dan reposisi bilangan-bilangan dalam array yang lebih dari 

bilangan yang dipilih di sisi yang lain. 

 Sisipkan bilangan yang dipilih di antara kedua kelompok ini. 

 

Di akhir partisi, belum semua array terurut, karena bilangan yang kurang atau lebih dari 

bilangan yang dipilih masih seperti semula. Prosedur ini baru membagi array ke dalam 

dua array yang lebih kecil yang lebih mudah untuk diurutkan 

 Prosedur partition menggunakan array untuk memasukkan list unsur dalam array 

dapat diubah langsung. Prosedur quicksort lebih mudah untuk dipahami jika terlebih dulu 

memahami prosedur partition. 

 

> partition:=proc(A::array(1,numeric),m::posint,n::posint) 

>   i:=m; 

>   j:=n; 

>   x:=A[j]; 

>   while i<j do 

>     if A[i]>x then 

>       A[j]:=A[i]; 

>       j:=j-1; 

>       A[i]:=A[j]; 

>     else 

>       i:=i+1; 

>     end if; 

>   end do; 

>   A[j]:=x; 

>   eval(A); 

> end proc: 
Warning, `i` is implicitly declared local to procedure `partition` 

 

Warning, `j` is implicitly declared local to procedure `partition` 

 

Warning, `x` is implicitly declared local to procedure `partition` 

 

Maple menyatakan i, j, dan x local karena prosedur partition menyatakannya secara 

eksplisit. Prosedur partition juga secara eksplisit menyatakannya untuk A, tetapi A adalah 

parameter, bukan variable lokal. Karena tidak ada pernyataaan untuk nama eval, Maple 

menjadikannya nama global yang mengacu pada perintah eval. 

 Setelah mempartisi array berikut, semua unsur yang kurang dari 3 disimpan 

sebelum 3, dan unsur yang lebih dari 3 disimpan setelah 3. 

> a:=array([2,4,1,5,3]); 
 := a [ ], , , ,2 4 1 5 3  

> partition(a,1,5); 
[ ], , , ,2 1 3 5 4  

> eval(a); 
[ ], , , ,2 1 3 5 4  

  

Tahap akhir dalam prosedur qick-sort adalah menyisipkan prosedur partition ke 

dalam prosedur yang lebih luar. Pertama prosedur yang lebih luar mendefinisikan 

subprosedur partition, lalu mempartisi array.  

 

 

> quicksort:=proc(A::array(1,numeric),m::integer,n::integer) 

>   local partition, p; 

>   partition:=proc(m,n) 

>     i:=m; 

>     j:=n; x:=A[j]; 

>     while i<j do 

>       if A[i]>x then 

>         A[j]:=A[i]; 

>         j:=j-1; 

>         A[i]:=A[j]; 

>       else 

>         i:=i+1; 

>       end if; 

>     end do; 

>     A[j]:=x; 

>     p:=j; 

>   end proc: 

>   if m<n then 

>     p:=partition(m,n); 

>     quicksort(A,m,p-1); 

>     quicksort(A,p+1,n); 

>   end if; 

>   eval(A); 

> end proc: 

>  
Warning, `i` is implicitly declared local to procedure `partition` 

 

Warning, `j` is implicitly declared local to procedure `partition` 

 

Warning, `x` is implicitly declared local to procedure `partition` 

> a:=array([2,4,1,5,3]); 
 := a [ ], , , ,2 4 1 5 3  

> quicksort(a,1,5); 
[ ], , , ,1 2 3 4 5  

> eval(a); 
[ ], , , ,1 2 3 4 5  

 Maple menentukan variabel A dan p dalam subprosedur partition didefinisikan 

oleh parameter dan variabel lokal dari prosedur quicksort yang lebih luar.  

 

6. TUGAS KELOMPOK 

 Diskusikan mengapa nest2 tidak memberikan hasil sepert nest yang sebelumnya! 

 Buat prosedur sederhana yang menggunakan variabel lokal, variabel global, 

prosedur bersarang. 


